INTRODUZIONE ALL'AMBIENTE DI SVILUPPO

DI VISUAL BASIC .NET (BETA 2)

PREMESSA

Ogni applicativo (Windows Application) Visual Basic si compone, in generale, di una o più finestre (form) contenenti oggetti (control).

DESCRIZIONE GENERALE DELL'AMBIENTE DI SVILUPPO
Fornendo il comando File | New | Project dall’ambiente di sviluppo comparirà una finestra di dialogo nella quale occorrerà la prima volta (quando si crea un nuovo progetto) scegliere, per i nostri scopi, Windows Application, quindi fornire un nome ed eventualmente una posizione e infine premere Ok.

L'ambiente di sviluppo di Visual Basic .NET si presenta composto da una serie di "finestre" che dipendono della finestra principale dell’ambiente di sviluppo. In un certo istante possono essere visibili solo alcune finestre e non altre: è possibile richiamare e fare scomparire le finestre usando in maniera opportuna i comandi dell'ambiente di sviluppo stesso (menù View).

La finestra principale, che chiameremo finestra di design, sarà sempre aperta in quanto consente di attivare tutte le funzionalità dell'ambiente di sviluppo.

Per i nostri scopi sono poi di interesse le seguenti finestre che più oltre descriveremo in maggior dettaglio:

· la finestra Toolbox che contiene i bottoni per creare gli oggetti (controls) da disporre sulle finestre (form) dell'applicativo; in condizioni normali questa finestra compare e scompare a tendina dal bordo sinistro;

· la finestra Solution Explorer che contiene l'elenco dei componenti dell'applicativo corrente (finestre, moduli, references, ecc.);

· la finestra Properties che fornisce e consente di modificare le proprietà dell'oggetto (form o control) attualmente selezionato;

· nell’area cliente della finestra di design saranno presenti una o più finestre Form che contengono il modello delle finestre dell'applicativo che si intende progettare; sempre nell’area cliente della finestra di design saranno visualizzate una o più finestre Code che consentono di visualizzare, inserire, modificare il codice Visual Basic associato agli eventi sugli oggetti; per accedere alla diverse finestre presenti nell’area cliente della finestra di design sono presenti opportune linguette.

LA FINESTRA "DESIGN"
Tale finestra è costituita dall'usuale barra del titolo, da una barra dei menu e da una toolbar costituita da una serie di bottoni per richiamare le funzionalità di uso più frequente del menù. Il resto della finestra è dedicato a contenere le altre finestre di cui si è parlato in precedenza.

Elenchiamo qui di seguito, con una breve descrizione, le funzionalità del menù che sono di interesse per i nostri scopi:

SYMBOL 183 \f "Symbol" \s 10 \h
File | New | Project: consente di creare un nuovo progetto (applicativo) partendo dalla situazione iniziale. Specificare Windows Application nella successiva finestra di dialogo.

SYMBOL 183 \f "Symbol" \s 10 \h
File | Open Solution: consente di aprire un precedente progetto (applicativo) per poterlo modificare ed eseguire. I file di progetto (soluzione) hanno di solito l'estensione .sln.

SYMBOL 183 \f "Symbol" \s 10 \h
File | Save All: consente di salvare sul file corrente (di solito .sln) il progetto (applicativo) corrente e tutti i file connessi, accogliendo tutte le modifiche apportate.

SYMBOL 183 \f "Symbol" \s 10 \h
File | Save ...: salva su file le modifiche apportate al componente (form o modulo) correntemente selezionato nella finestra Solution Explorer.

SYMBOL 183 \f "Symbol" \s 10 \h
File | Save ... As: consente di scegliere un file su cui salvare il componente (form o modulo) correntemente selezionato nella finestra Solution Explorer.

SYMBOL 183 \f "Symbol" \s 10 \h
File | Print: consente di stampare il contenuto della finestra correntemente selezionata nell’area cliente della finestra di design.

SYMBOL 183 \f "Symbol" \s 10 \h
File | Exit: consente di uscire dall’ambiente di sviluppo del Visual Basic.

SYMBOL 183 \f "Symbol" \s 10 \h
Project | Add Windows Form: crea una nuova form (finestra) per il progetto (applicativo) attuale. Specificare Windows Form nella successiva finestra di dialogo.

SYMBOL 183 \f "Symbol" \s 10 \h
Project | Add Module: crea un nuovo modulo per il progetto (applicativo) attuale. Specificare Module nella finestra di dialogo successiva.

SYMBOL 183 \f "Symbol" \s 10 \h
Project | Add Existing Item: aggiunge al progetto corrente un componente (form, modulo, ecc.) salvato in precedenza.

SYMBOL 183 \f "Symbol" \s 10 \h
Project / Exclude From Project: elimina dal progetto corrente (senza distruggerlo) il componente (form, modulo, ecc.) attualmente selezionato nella finestra Solution Explorer.

SYMBOL 183 \f "Symbol" \s 10 \h
Project | Property Pages  (quando è selezionato un progetto in Solution Explorer): consente di specificare alcune opzioni del progetto corrente. Di particolare interesse è la possibilità di definire o cambiare la finestra di "partenza" dell'applicativo (Start Up Object) all’interno di Common Properties | General voce Startup Object. La finestra di "partenza" è quella che viene visualizzata inizialmente all'atto dell'esecuzione dell'applicativo.

SYMBOL 183 \f "Symbol" \s 10 \h
Project | Add Inherited control: aggiunge al progetto corrente un nuovo componente (controllo) da scegliere andando a ricercare la dll che contiene il controllo medesimo.

SYMBOL 183 \f "Symbol" \s 10 \h
Edit | Cut: elimina l'oggetto (control) o la porzione di testo selezionata e la trasferisce nella clipboard (appunti).

SYMBOL 183 \f "Symbol" \s 10 \h
Edit | Copy: copia l'oggetto (control) o la porzione di testo selezionata e la mette nella clipboard (appunti).

SYMBOL 183 \f "Symbol" \s 10 \h
Edit | Paste: inserisce l'oggetto (control) o il testo presente nella clipboard (appunti) rispettivamente nella form corrente o nella finestra di codice corrente (alla posizione del cursore).

SYMBOL 183 \f "Symbol" \s 10 \h
Edit | Delete: elimina l'oggetto (control) o la porzione di testo selezionata ma non la trasferisce nella clipboard (appunti).

SYMBOL 183 \f "Symbol" \s 10 \h
Edit | Find And replace | Find: cerca una "stringa" all'interno del codice del progetto corrente.

· Edit | Find And replace | Replace: cerca una "stringa" all'interno del codice del progetto corrente e la sostituisce con una nuova "stringa" data.

SYMBOL 183 \f "Symbol" \s 10 \h
Build | Build: consente di creare un applicativo Windows (file EXE) "indipendente" a partire dal progetto attuale.

SYMBOL 183 \f "Symbol" \s 10 \h
Debug | Start: consente di eseguire l'applicativo (progetto) attuale.

SYMBOL 183 \f "Symbol" \s 10 \h
Debug | Stop Debugging: consente di terminare l'esecuzione dell'applicativo (progetto) attuale.

SYMBOL 183 \f "Symbol" \s 10 \h
Window: consente di attivare e portare in primo piano una finestra.

SYMBOL 183 \f "Symbol" \s 10 \h
View: consente di attivare una finestra tra quelle menzionate in precedenza per l’ambiente di sviluppo. In particolare è possibile attivare le seguenti finestre: la finestra Toolbox (View | Toolbox); la finestra Solution Explorer (View | Solution Explorer); la finestra Properties (View | Properties Window); la finestra Code (View | Code); una finestra Form (quella selezionata) con il comando View | Designer.

SYMBOL 183 \f "Symbol" \s 10 \h
Help: consente l'accesso ad una comoda e potente guida in linea sul Visual Basic e sul suo ambiente di sviluppo. 

LA FINESTRA "TOOLBOX"
La finestra Toolbox contiene gli oggetti che è possibile inserire su di un form. Se non si aggiungono componenti, la Toolbox conterrà solo gli elementi standard. In tale situazione, elenchiamo, gli oggetti (controls) che è possibile disporre su di un form e che sono di interesse per noi (nella categoria Windows Forms):

1. Pointer per "selezionare" un oggetto sulla form (finestra).

2. Label per un oggetto di testo “statico” (etichetta)

3. Button per un pulsante
4. TextBox per una casella di testo
5. MainMenu per il menù principale
6. CheckBox per un pulsante di opzione (a scelta multipla)
7. RadioButton per un pulsante di opzione (a scelta singola)
8. GroupBox per creare un gruppo di controlli
9. PictureBox per una immagine bitmap
10. ListBox per una casella di riepilogo
11. ComboBox per una casella di riepilogo a discesa
12. HscrollBar per una barra di scorrimento orizzontale
13. VscrollBar per una barra di scorrimento verticale
14. Timer per un oggetto timer
15. ContextMenu per un menù shortcut
Per disporre su di una form ciascuno dei precedenti oggetti occorre:

· Fare comparire la tolbox;

· eseguire un drag sulla form di interesse per disporre e quindi dimensionare l'oggetto.

LA FINESTRA "SOLUTION EXPLORER"
La finestra Solution Explorer consente di visualizzare i componenti di una applicazione (progetto) Visual Basic e di passare da un componente all'altro. La dove ha senso, una volta selezionato un componente sulla finestra Solution Explorer, è possibile visualizzare il codice (View | Code) o la finestra (View | Designer) associati.

LE FINESTRE "FORM"
Le finestre Form sono quelle in cui si disegna effettivamente l'interfaccia utente dell'applicativo Visual Basic che si vuole costruire. Le tipiche operazioni che si possono eseguire su di una finestra Form sono, per quanto ci può interessare, le seguenti:

· Fare click su un oggetto (control o form) per selezionarlo.

· Fare doppio click su un oggetto (control o form) per aprire la finestra di codice che consente di "editare" il codice "associato" agli eventi sull'oggetto.

· Premere il tasto F4 dopo avere selezionato un oggetto (control o form) per fare comparire la finestra Properties associata a quell'oggetto.

· Eseguire una operazione di drag con il mouse per spostare un oggetto (control o form) in una diversa posizione.

· Eseguire una operazione di drag con il mouse sui marker di un control o sui marker di una finestra (form) per ridimensionare l'oggetto (control o form).

· Premere il tasto CANC per cancellare l'oggetto (control) attualmente selezionato.

LA FINESTRA "PROPERTIES"
Tale finestra contiene:

· Un combo box per la scelta dell'oggetto; per default viene proposto l'oggetto selezionato.

· Una listbox per la visualizzazione e la scelta delle coppie proprietà-valore dell'oggetto attuale. Esistono in realtà due bottoni per consentire di accedere alle coppie proprietà-valore in ordine alfabetico o in ordine di “categoria”.

· Un control per modificare il valore della proprietà corrente dell'oggetto attuale: questo control è rappresentato dallo spazio a destra del nome della proprietà.

LA FINESTRA "CODE"
Tale finestra contiene principalmente i seguenti elementi:

· Un combo box per la scelta dell'oggetto ai cui eventi si vuole associare il codice Visual Basic. E' importante osservare che oltre alla lista degli oggetti esiste sempre una voce (che ha di solito il nome del form) che consente di definire il codice per la dichiarazione delle variabili globali alla form e per la definizione delle funzioni e procedure globali alla form. Esiste anche una voce Overrides per ridefinire i metodi della classe base (Windows Forms). Esiste infine la voce Base Class Events che contiene gli eventi per la classe base (la classe Windows Forms) che consente di accedere a tutti gli eventi di base dei form.

· Un combo box per la scelta dell'evento relativo all'oggetto a cui si vuole associare il codice Visual Basic. Quando nel precedente combo box è inserita la voce “Form …” in questo combo box è possibile scegliere la voce "(Declaration)" o una delle funzioni o procedure scritte in precedenza. Nuove procedure o funzioni possono essere aggiunte sotto la voce "(Declaration)".

· L'area cliente contenente lo spazio per la scrittura e modifica del codice Visual Basic associato all'evento e all'oggetto selezionati nei due combo box precedenti. Di solito una volta scelti oggetto ed evento l'area cliente presenta già una "intestazione" che si riferisce proprio alla coppia oggetto-evento selezionata. La scrittura e la modifica del codice avviene in maniera usuale, semplice e naturale e non merita perciò nessun commento particolare.

NOTE CONCLUSIVE
Ogni finestra dell'ambiente di sviluppo Visual Basic può essere chiusa usando il box di chiusura (a forma di “X”).

In condizioni normali è abilitato il controllo sintattico sulle righe di codice che si scrivono o modificano e perciò si può immediatamente riconoscere un errore di sintassi dalla presenza di una “ondina” sotto l’istruzione errata.

DISPENSA INTRODUTTIVA AL LINGUAGGIO VISUAL BASIC .NET
Per chi conosce linguaggi di programmazione “tradizionali” l’approccio alla programmazione in Windows richiede alcune premesse fondamentali che sono riassumibili nei seguenti concetti:

1. Il concetto di oggetto: in ambiente DOS l’unico tipo di oggetto che si incontra a livello fondamentale è il carattere (l’input e l’output dei programmi è costituito da sequenze di caratteri). Basta una occhiata a Windows per rendersi conto che le cose sono diverse: in una videata sotto Windows possiamo trovare diversi tipi di oggetti come finestre, bottoni, edit-control, static-control, listbox, combobox, checkbox, radiobutton, testo, immagini bitmap ed altri ancora. Sembra allora ovvio che la programmazione in Windows non possa prescindere dalla presenza e dall’uso di tali oggetti.

2. Il concetto di evento (su di un oggetto): in DOS ogni programma richiama le proprie funzionalità a seconda dei dati di ingresso; è quindi sotto il controllo del programma che avviene il richiamo delle diverse funzionalità del programma medesimo. In Windows un programma è in genere suddiviso in tante “routines” ciascuna delle quali gestisce un particolare evento su di un oggetto. Queste “routines” non vengono tipicamente eseguite su richiesta del “programma” ma vengono mandate in esecuzione da Windows stesso ogni volta che si verifica quel dato evento su quel dato oggetto. Esempi di eventi sono il click del mouse (su una finestra, su un bottone, su un radiobutton, su un edit-control, su una listbox, ...), il doppio click del mouse (su una finestra, su un bottone, su un radiobutton, su un edit-control, su una listbox, ...), l’apertura di una finestra, la pressione di un tasto (su un edit-control, su un combobox, ...), il resize di una finestra, eccetera.

3. Il concetto di proprietà di un oggetto: le proprietà sono le caratteristiche di un oggetto. E’ possibile leggere e/o cambiare le proprietà di un oggetto: il testo di un edit-control, il testo di uno static-control, il colore di sfondo di una finestra, il titolo di una finestra, l’abilitazione di un checkbox, eccetera.

4. Il concetto di metodo associato ad un oggetto: i metodi sono gli operatori che consentono di lavorare su di un oggetto. Per passare “il controllo” (focus) ad un edit-control di nome “Edit1” si userà il metodo Edit1.Select.

Da quanto detto sopra nasce spontanea la seguente metodologia per la costruzione di applicativi sotto Windows:

1. Diagramma di flusso dei dati: principi e caratteristiche sono le stesse dei diagrammi di flusso associati a programmi DOS.

2. Tracciati record e tracciati di stampa: principi e caratteristiche  sono le stesse  dei tracciati corrispondenti a programmi sotto DOS. Un esempio di tracciato di stampa è sviluppato all’interno del successivo punto 5.

3. Tracciato video: la progettazione dei formati video per applicativi Windows è diversa da quella per applicativi DOS. In windows le videate (finestre) indicano la struttura dell’applicativo e condizionano in maniera molto pesante tutto il successivo sviluppo: sono quindi elementi fondamentali della progettazione. Per tale motivo, spesso, gli ambienti di programmazione Windows forniscono strumenti per la costruzione delle finestre (come in Visual Basic). La progettazione e costruzione delle finestra consente di individuare ed elencare gli oggetti di cui si è parlato in precedenza. Vediamo un esempio.

Finestre per la determinazione e stampa di una fattura (versione semplificata):

Elenco degli oggetti delle finestre mostrate nella precedente figura:

· Finestra “Fatturazione”

· Group box “Valore della merce”

· Label (Static-control) “Quantità merce”

· Textbox (Edit-control) per la quantità di merce

· Label “Prezzo merce”

· Textbox per il prezzo unitario della merce

· Label “Importo merce”

· Label per contenere l’importo della merce

· Label “Importo progr. merce”

· Label per contenere l’importo progressivo della merce

· Bottone “Aggiungi”

· Bottone “Stampa”

· Bottone “Esci”

· Finestra “Maschera di stampa”

· Bottone “Stampa”

4. Individuazione degli eventi sugli oggetti: occorre individuare tra gli eventi possibili quelli a cui si desidera associare una azione diversa da quella di default di Windows. Per la programmazione in Windows questa è una parte fondamentale. Anche qui possiamo chiarire la cosa riprendendo l’esempio precedente.

Eventi relativi alle finestre: dovendo le finestre consentire di compilare ed eventualmente stampare una fattura partendo dai dati inseriti dall’utente nella maschera, gli eventi a cui è necessario associare una azione sono i seguenti:

· Evento “Creazione della finestra” per la finestra “Fatturazione” (Sub New): si verifica all’atto del caricamento della finestra.

· Evento "Click su Bottone Aggiungi" per la finestra “Fatturazione”.

· Evento "Click su Bottone Stampa" per la finestra “Fatturazione”.

· Evento "Click su Bottone Esci" per la finestra “Fatturazione”.
· Evento “Chiusura della finestra” per la finestra “Maschera di stampa”
· Evento “Click su Bottone Stampa” per la finestra “Maschera di stampa”
· Evento “Stampa pagina” per la finestra “Maschera di stampa”: accade ogni volta che deve essere stampata una pagina.
5. Per ognuno degli eventi individuati in precedenza si costruisce “l’analisi” dell’azione da svolgere: occorre, in sostanza, descrivere cosa occorre fare in risposta all’evento. Questa analisi non differisce in alcun modo, dal punto di vista concettuale, da quella che si può fare per applicativi in ambiente DOS, anche se qui si potrà fare riferimento ai concetti propri di Windows esposti all’inizio (oggetti, proprietà, metodi, eventi). Riprendendo l’esempio precedente proponiamo l’analisi delle risposte agli eventi usando come strumento la pseudocodifica.

· Evento “Creazione della finestra” per la finestra “Fatturazione” (Sub New):

BEGIN
  <si crea un riferimento alla form per la stampa della fattura>

  <si inizializza una fattura vuota>

END
· Evento "Click su Bottone Aggiungi" per la finestra “Fatturazione”:

BEGIN
  <lettura e memorizzazione di quantità e prezzo unitario dai relativi 

   Textbox>

  <determinazione e memorizzazione dell’importo delle merce attuale>

  <totalizzazione dell’importo attuale con il totale precedente>

  <visualizzazione negli appositi Label dell’importo 

   attuale e di quello progressivo della merce>

  <aggiornamento del numero di merci>

END
· Evento "Click su Bottone Stampa" per la finestra “Fatturazione”.

BEGIN

    <visualizzo la maschera di stampa>

    <nascondo la maschera di Fatturazione>

END

· Evento "Click su Bottone Esci" per la finestra “Fatturazione”.
BEGIN
  <Terminazione del programma>

END

· Evento “Chiusura della finestra” per la finestra “Maschera di stampa”
BEGIN
  <nascondo la finestra “Stampa”>

  <visualizzo la finestra “Fatturazione”>

END

· Evento “Click su Bottone Stampa” per la finestra “Maschera di stampa”
BEGIN
  <faccio partire la stampa>

END

· Evento “Stampa pagina” per la finestra “Maschera di stampa” : si fa riferimento al seguente tracciato stampa:

FATTURA

-------------------------------

N.    QUANTITA’     PREZZO     IMPORTO

999  99999999999  9999999999  999999999

 :        :           :           :

 :        :           :           :

999  99999999999  9999999999  999999999

                            -----------

    TOTALE ...............    999999999

-------------------------------

BEGIN

  <inizializzazione delle operazioni di stampa>

  <stampa delle 3 righe di intestazione>

  FOR <ogni merce caricata> DO
    <stampa una riga con: numero progressivo, quantità della merce,

     prezzo unitario della merce, importo della merce>

  ENDDO

  <stampa delle 3 righe di conclusione con il totale>

  <termina la stampa>

END
6. Si procede ora alla codifica in linguaggio Visual BASIC delle “risposte” agli eventi sopra individuate. La codifica avviene “traducendo” le descrizioni del precedente punto 5 in termini di istruzioni del linguaggio. All’inizio vanno premesse le dichiarazioni delle variabili globali che occorrono al programma. Per mostrare la situazione riprendiamo l’esempio precedente.

· Dichiarazioni globali nella finestra “Fatturazione”:
    Private frmForm2 As New Form2()

    Private pos As Integer

La prima istruzione dichiara e istanzia una variabile che conterrà il riferimento alla finestra “Maschera di stampa”.

· Dichiarazioni globali nella finestra “Maschera di stampa”:
    Private WithEvents Printer As Printing.PrintDocument

    Public objForm1 As Form1

    Public aPrezzo(100) As Long

    Public aQuant(100) As Long

    Public aImpo(100) As Long

    Public ProgImpo As Long

    Public nElem As Integer

La prima istruzione dichiara l’oggetto stampante (con eventi) per la gestione della stampa. La seconda istruzione dichiara una variabile che conterrà il riferimento alla finestra “Fatturazione”.

· Evento “Creazione della finestra” per la finestra “Fatturazione” (Sub New):

    Public Sub New()

        MyBase.New

        Form1 = Me

        'This call is required by the Win Form Designer.

        InitializeComponent

        'TODO: Add any initialization after the InitializeComponent() call

        frmForm2.objForm1 = Me

        pos = 0

    End Sub

Le istruzioni di interesse sono quelle dopo il commento “TODO”, le altre sono automaticamente generate dall’ambiente di sviluppo. La prima di tali istruzioni pone un riferimento al form corrente (form “Fatturazione”) nella opportuna proprietà del form “Maschera di stampa”).

· Evento "Click su Bottone Aggiungi" per la finestra “Fatturazione”:

    Private Sub Button3_Click(ByVal sender As System.Object, _

                              ByVal e As System.EventArgs) _

            Handles Button3.Click

        Dim tmp As Long

        frmForm2.aQuant(pos) = TextBox1.Text

        frmForm2.aPrezzo(pos) = TextBox2.Text

        frmForm2.aImpo(pos) = frmForm2.aQuant(pos) * frmForm2.aPrezzo(pos)

        frmForm2.ProgImpo = frmForm2.ProgImpo + frmForm2.aImpo(pos)

        Importo.Text = frmForm2.aImpo(pos).ToString

        tmp = frmForm2.ProgImpo

        ImportoProg.Text = tmp.ToString

        pos = pos + 1

        frmForm2.nElem = pos

    End Sub

· Evento "Click su Bottone Stampa" per la finestra “Fatturazione”.

    Private Sub Button1_Click(ByVal sender As System.Object, _

                              ByVal e As System.EventArgs) _

            Handles Button1.Click

        frmForm2.Show()

        Me.Hide()

    End Sub

· Evento "Click su Bottone Esci" per la finestra “Fatturazione”.
    Private Sub Button2_Click(ByVal sender As System.Object, _

                              ByVal e As System.EventArgs) _

            Handles Button2.Click

        End

    End Sub

· Evento “Chiusura della finestra” per la finestra “Maschera di stampa”
    Public Sub Form2_Closing(ByVal sender As Object, _

                             ByVal e As System.ComponentModel.CancelEventArgs) _   

           Handles Form2.Closing

        e.Cancel = True

        Me.Hide()

        objForm1.Show()

    End Sub

· Evento “Click su Bottone Stampa” per la finestra “Maschera di stampa”
    Private Sub Button1_Click(ByVal sender As System.Object, _

                              ByVal e As System.EventArgs) _

            Handles Button1.Click

        Printer = New Printing.PrintDocument()

        Printer.Print()

    End Sub

· Evento “Stampa pagina” per la finestra “Maschera di stampa”
    Public Sub Printer_PrintPage(ByVal sender As Object, _

                   ByVal e As System.Drawing.Printing.PrintPageEventArgs) _

           Handles Printer.PrintPage

        Dim myFont As New Font("Courier New", 10, Drawing.FontStyle.Regular, _

                               Drawing.GraphicsUnit.Point)

        Dim Riga As String

        Dim Elem As String

        Dim XIn As Integer

        Dim YIn As Integer

        Dim Passo As Integer

        Dim I As Integer

        XIn = 10

        YIn = 10

        Passo = 20

        e.Graphics.DrawString("FATTURA", myFont, System.Drawing.Brushes.Black, _

                              XIn, YIn)

        Riga = "-------------------------------"

        YIn = YIn + Passo

        e.Graphics.DrawString(Riga, myFont, System.Drawing.Brushes.Black, _

                              XIn, YIn)

        Riga = "N.    QUANTITA'     PREZZO     IMPORTO "

        YIn = YIn + Passo

        e.Graphics.DrawString(Riga, myFont, System.Drawing.Brushes.Black, _

                              XIn, YIn)

        For I = 0 To (nElem - 1)

            Elem = I.ToString

            Elem = Space(3 - Elem.Length) & Elem

            Riga = Elem & "  "

            Elem = aQuant(I).ToString

            Elem = Space(11 - Elem.Length) & Elem

            Riga = Riga & Elem & "  "

            Elem = aPrezzo(I).ToString

            Elem = Space(10 - Elem.Length) & Elem

            Riga = Riga & Elem & "  "

            Elem = aImpo(I).ToString

            Elem = Space(9 - Elem.Length) & Elem

            Riga = Riga & Elem

            YIn = YIn + Passo

            e.Graphics.DrawString(Riga, myFont, System.Drawing.Brushes.Black, _

                                  XIn, YIn)

        Next

        Riga = "                            -----------"

        YIn = YIn + Passo

        e.Graphics.DrawString(Riga, myFont, System.Drawing.Brushes.Black, _

                              XIn, YIn)

        Elem = ProgImpo.ToString

        Elem = Space(9 - Elem.Length) & Elem

        Riga = "    TOTALE ...............    " & Elem

        YIn = YIn + Passo

        e.Graphics.DrawString(Riga, myFont, System.Drawing.Brushes.Black, _

                              XIn, YIn)

        Riga = "-------------------------------"

        YIn = YIn + Passo

        e.Graphics.DrawString(Riga, myFont, System.Drawing.Brushes.Black, _

                              XIn, YIn)

        e.HasMorePages = False

    End Sub

 ACCESSO A DATABASE IN AMBIENTE VISUAL BASIC .NET (BETA 2)

1. DEFINIZIONI GENERALI
Una modalità di accesso a database è quella che passa attraverso OleDB. Limitandoci ai database Access, possiamo definire più specificatamente i tipi di oggetti che sono necessari per potere accedere al database.

Per operare da programma sugli oggetti del database (di solito tabelle o query) occorre avere a disposizione i seguenti tipi di oggetti Visual Basic .Net:

· Oggetto di tipo System.Data.OleDb.OleDBConnection che rappresenta la connessione con il database nel suo complesso.

· Oggetto di tipo System.Data.OleDb.OleDbCommand che rappresenta un comando (es. SELECT di SQL) inviato ad un database.

· Oggetto di tipo System.Data.OleDb.OleDbDataReader che rappresenta recordset (tabella) vista in sola lettura.

· Oggetto di tipo System.Data.OleDb.OleDbDataAdapter che rappresenta l’oggetto in grado di popolare un DataSet (vedi sotto)  partire da un comando e viceversa di aggiornare il database a partire dalle modifiche operate su un DataSet (vedi sotto).

· Oggetto di tipo System.Data.OleDb.OleDbCommandBuilder che rappresenta l’oggetto in grado generare in automatico le istruzioni SQL necessarie per l’aggiornamento del database attraverso un DataAdapter (vedi sopra).

· Oggetto di tipo System.Data.Dataset che rappresenta un insieme di recordset (tabelle e query) con le relative relazioni. Di solito avremo dataset contenenti un solo recordset (tabella o query).

· Oggetto di tipo System.Data.DataRow che rappresenta una riga di un recordset (tabella).

2. UN ESEMPIO DI PROGRAMMA CHE USA DATABASE
Allo scopo di mostrare caratteristiche ed utilizzo degli oggetti e dei concetti discussi al paragrafo precedente si fornisce un esempio, completo di documentazione, di un programma che gestisce un archivio anagrafico.

2.1 Diagramma di flusso








2.2 Tracciato record

Database: FLANAG.MDB

Tabella: FLANAG


Indice su ID.

2.3 Tracciato Video


[image: image3.png]~=lolx|

Mostia

Aggioma

Cancela

0: | Indiizzo

L —

anag: [ Data. |

ndiens [





2.4 Elenco eventi

SYMBOL 183 \f "Symbol" \s 10 \h
Click su bottone "Mostra"

SYMBOL 183 \f "Symbol" \s 10 \h
Click su bottone "Aggiorna"

SYMBOL 183 \f "Symbol" \s 10 \h
Click su bottone "Cancella"

SYMBOL 183 \f "Symbol" \s 10 \h
Click su bottone "Inserisci"

2.5 Analisi delle risposte agli eventi

SYMBOL 183 \f "Symbol" \s 10 \h
Click su bottone "Mostra"

BEGIN

  <mi connetto a database>

  <reperisco l’elenco anagrafico di tutti i soggetti>

  <pulisco la listbox dei soggetti>

  WHILE <l’elenco dei soggetti non è terminato> DO

    <aggiungo il soggetto attuale alla listbox>

    <passo al prossimo soggetto>

  END WHILE

END

SYMBOL 183 \f "Symbol" \s 10 \h
Click su bottone "Aggiorna"

BEGIN

  <mi connetto a database>

  <predispongo le strutture necessarie per il successivo reperimento e aggiornamento dei dati>

  <reperisco l’elenco anagrafico di tutti i soggetti>

  FOR <ogni soggetto dell’elenco anagrafico> DO

    IF <l’identificativo del soggetto attuale è uguale a quello specificato

       nel textbox dall’utente> THEN

       <memorizzo il soggetto attuale>

     END IF

  END FOR

  IF <è stato trovato il soggetto ricercato> THEN

    <modifico l’anagrafica del soggetto trovato>

    <riporto le modifiche sul database>

  ELSE

    <emetto un messaggio di errore>

  END IF

END

SYMBOL 183 \f "Symbol" \s 10 \h
Click su bottone "Cancella"

BEGIN

  <mi connetto a database>

  <predispongo le strutture necessarie per il successivo reperimento e aggiornamento dei dati>

  <reperisco l’elenco anagrafico di tutti i soggetti>

  FOR <ogni soggetto dell’elenco anagrafico> DO

    IF <l’identificativo del soggetto attuale è uguale a quello specificato

       nel textbox dall’utente> THEN

       <memorizzo il soggetto attuale>

     END IF

  END FOR

  IF <è stato trovato il soggetto ricercato> THEN

    <elimino il soggetto trovato>

    <riporto le modifiche sul database>

  ELSE

    <emetto un messaggio di errore>

  END IF

END

SYMBOL 183 \f "Symbol" \s 10 \h
Click su bottone "Inserisci"

BEGIN

  <mi connetto a database>

  <predispongo le strutture necessarie per il successivo reperimento e aggiornamento dei dati>

  <reperisco l’elenco anagrafico di tutti i soggetti>

  <creo un nuovo soggetto>

  <carico i dati letti dai textbox nel nuovo soggetto>

  <aggiungo il nuovo soggetto all’elenco dei soggetti>

  <salvo i dati sul database>

END

2.6 Codice Visual Basic del programma.

Presentiamo qui di seguito la codifica del programma Visual Basic (WindowsApplication1) nelle sue diverse componenti. Alla fine di ogni componente sono posti commenti che consentono di spiegare l'uso delle diverse istruzioni coinvolte.

SYMBOL 183 \f "Symbol" \s 10 \h
Dichiarazioni generali alla classe:

    Public conn As System.Data.OleDb.OleDbConnection

    Public cmd As System.Data.OleDb.OleDbCommand

    Public rs As System.Data.OleDb.OleDbDataReader

SYMBOL 183 \f "Symbol" \s 10 \h Click su bottone "Mostra"

    Private Sub Button1_Click(ByVal sender As System.Object, _

                              ByVal e As System.EventArgs) _

            Handles Button1.Click

        Dim strdati As String

        Dim cmd As Data.OleDb.OleDbCommand

        Dim rs As Data.OleDb.OleDbDataReader

        Dim strconn As String

        strconn = "Provider=Microsoft.Jet.OLEDB.4.0;"

        strconn = strconn & "Data Source=C:\Giovanni\FlAnag2000.mdb;"

        strconn = strconn & "Persist Security Info=False;"

        conn = New Data.OleDb.OleDbConnection()

        cmd = New Data.OleDb.OleDbCommand()

        conn.ConnectionString = strconn

        conn.Open()

        cmd.CommandText = "SELECT * FROM FLANAG"

        cmd.Connection = conn

        rs = cmd.ExecuteReader()

        ListBox1.Items.Clear()

        While rs.Read

            strdati = CStr(rs("ID")) & ") " & CStr(rs("ANAGRAF")) & _

                      ": " & CStr(rs("INDIRIZZO"))

            ListBox1.Items.Add(strdati)

        End While

    End Sub

1. Nella ConnectionString (stringa di connessione) viene precisato il driver (provider) per Access (JET), il pathname del database e altre eventuali informazioni. Di seguito viene aperta la connessione con il database (metodo Open).

2. Il CommandText contiene l’istruzione SQL da eseguire. In questo caso “SELECT * FROM FLANAG” seleziona (nel recordset risultante) tutti i campi di tutti i record di FLANAG. Di seguito viene utilizzata la connessione del precedente per accedere al database (proprietà Connection). Infine viene eseguito il comando popolando il recordset (DataReader) rs (metodo ExecuteReader).

3. Il metodo Read dell’oggetto rs (recordset) consente di passare al record successivo: il valore di ritorno dice se sono arrivato alla fine del file.

4. L’espressione rs(“NOME_CAMPO”) consente di accedere al valore dal campo (che è la proprietà di default).

SYMBOL 183 \f "Symbol" \s 10 \h
Click su bottone "Aggiorna"

    Private Sub Button2_Click(ByVal sender As System.Object, _

                              ByVal e As System.EventArgs) _

            Handles Button2.Click

        Dim myCmd As Data.OleDb.OleDbCommand

        Dim myDA As Data.OleDb.OleDbDataAdapter

        Dim myCB As Data.OleDb.OleDbCommandBuilder

        Dim myDS As DataSet

        Dim myDS1 As DataSet

        Dim myRow As DataRow

        Dim curRow As DataRow

        Dim strconn As String

        Dim strSQL As String

        strconn = "Provider=Microsoft.Jet.OLEDB.4.0;"

        strconn = strconn & "Data Source=C:\Giovanni\FlAnag2000.mdb;"

        strconn = strconn & "Persist Security Info=False;"

        strSQL = "SELECT * FROM FLANAG"

        conn = New Data.OleDb.OleDbConnection(strconn)

        myDS = New System.Data.DataSet()

        myCmd = New Data.OleDb.OleDbCommand(strSQL, conn)

        myDA = New Data.OleDb.OleDbDataAdapter(myCmd)

        myCB = New Data.OleDb.OleDbCommandBuilder(myDA)

        myDA.Fill(myDS, "FLANAG")

        myRow = Nothing

        For Each curRow In myDS.Tables(0).Rows

            If CStr(curRow("ID")) = TextBox1.Text Then

                myRow = curRow

            End If

        Next curRow

        If Not (IsNothing(myRow)) Then

            myRow("INDIRIZZO") = TextBox2.Text

            myDS1 = myDS.GetChanges(Data.DataRowState.Modified)

            myDS1.Merge(myDS)

            myDA.Update(myDS, "FLANAG")

            myDS.AcceptChanges()

        Else

            MsgBox("ID inesistente", , "AVVISO")

        End If

    End Sub

1. Nelle variabili strconn e strSQL vengono preparati rispettivamente la stringa di connessione e l’istruzione SQL da eseguire (vedi evento precedente).

2. Viene predisposta la connessione al database e un oggetto Dataset per contenere il recordset su cui eseguire l’aggiornamento.

3. Viene creato un comando (OleDbCommand) che attraverso una connessione a database eseguirà l’istruzione SQL per la selezione del recordset su cui eseguire le modifiche. Nello stesso contesto viene creato il DataAdapter che si occuperà di reperire i dati da database ed eseguire le modifiche sul database. Nel DataAdapter vengono pure create le istruzioni necessarie per l’aggiornamento dei dati sul database usando un CommandBuilder.

4. Attraverso il metodo Fill viene eseguita l’istruzione SQL e viene popolato il Dataset myDS con i dati del recordset da modificare.

5. Il ciclo di For passa in rassegna tutti i record del recordset (elementi della collezione Rows) e se trova quello di interesse lo salva nella variabile curRow. L’eventuale record trovato è quello da modificare.

6. Se è stato trovato il record da modificare, si cambia valore al campo INDIRIZZO usando l’usuale sintassi di accesso al campo: rs(“INDIRIZZO”). A questo punto il salvataggio delle modifiche sul database richiede i seguenti passi:

a. Viene creato un dataset contenente solo i record modificati (uno solo) nel Dataset originale (myDS): metodo GetChanges.

b. Con il metodo Merge i cambiamenti del nuovo dataset vengono inclusi nel dataset originale (myDS).

c. Attraverso i metodi Update dell’oggetto DataAdapter e AcceptChanges del Dataset si trasferiscono effettivamente le modifiche del Dataset sul database.

SYMBOL 183 \f "Symbol" \s 10 \h
Click su bottone "Cancella"

    Private Sub Button3_Click(ByVal sender As System.Object, _

                              ByVal e As System.EventArgs) _

            Handles Button3.Click

        Dim myCmd As Data.OleDb.OleDbCommand

        Dim myDA As Data.OleDb.OleDbDataAdapter

        Dim myCB As Data.OleDb.OleDbCommandBuilder

        Dim myDS As DataSet

        Dim myDS1 As DataSet

        Dim myRow As DataRow

        Dim curRow As DataRow

        Dim strconn As String

        Dim strSQL As String

        strconn = "Provider=Microsoft.Jet.OLEDB.4.0;"

        strconn = strconn & "Data Source=C:\Giovanni\FlAnag2000.mdb;"

        strconn = strconn & "Persist Security Info=False;"

        strSQL = "SELECT * FROM FLANAG"

        conn = New Data.OleDb.OleDbConnection(strconn)

        myDS = New System.Data.DataSet()

        myCmd = New Data.OleDb.OleDbCommand(strSQL, conn)

        myDA = New Data.OleDb.OleDbDataAdapter(myCmd)

        myCB = New Data.OleDb.OleDbCommandBuilder(myDA)

        myDA.Fill(myDS, "FLANAG")

        myRow = Nothing

        For Each curRow In myDS.Tables(0).Rows

            If CStr(curRow("ID")) = TextBox3.Text Then

                myRow = curRow

            End If

        Next curRow

        If Not (IsNothing(myRow)) Then

            myRow.Delete()

            myDS1 = myDS.GetChanges(Data.DataRowState.Deleted)

            myDS1.Merge(myDS)

            myDA.Update(myDS, "FLANAG")

            myDS.AcceptChanges()

        Else

            MsgBox("ID inesistente", , "AVVISO")

        End If

    End Sub

1. Nelle variabili strconn e strSQL vengono preparati rispettivamente la stringa di connessione e l’istruzione SQL da eseguire (vedi evento precedente).

2. Viene predisposta la connessione al database e un oggetto Dataset per contenere il recordset su cui eseguire l’aggiornamento.

3. Viene creato un comando (OleDbCommand) che attraverso una connessione a database eseguirà l’istruzione SQL per la selezione del recordset su cui eseguire le modifiche. Nello stesso contesto viene creato il DataAdapter che si occuperà di reperire i dati da database ed eseguire le modifiche sul database. Nel DataAdapter vengono pure create le istruzioni necessarie per l’aggiornamento dei dati sul database usando un CommandBuilder.

4. Attraverso il metodo Fill viene eseguita l’istruzione SQL e viene popolato il Dataset myDS con i dati del recordset da modificare.

5. Il ciclo di For passa in rassegna tutti i record del recordset (elementi della collezione Rows) e se trova quello di interesse lo salva nella variabile curRow. L’eventuale record trovato è quello da cancellare.

6. Se è stato trovato il record da cancellare lo si elimina usando il metodo Delete. A questo punto il salvataggio delle modifiche sul database richiede i seguenti passi:

a. Viene creato un dataset contenente solo i record cancellati (uno solo) nel Dataset originale (myDS): metodo GetChanges.

b. Con il metodo Merge i cambiamenti del nuovo dataset vengono inclusi nel dataset originale (myDS).

c. Attraverso i metodi Update dell’oggetto DataAdapter e AcceptChanges del Dataset si trasferiscono effettivamente le modifiche del Dataset sul database.

SYMBOL 183 \f "Symbol" \s 10 \h
Click su bottone "Inserisci"

    Private Sub Button4_Click(ByVal sender As System.Object, _

                              ByVal e As System.EventArgs) _

            Handles Button4.Click

        Dim myCmd As Data.OleDb.OleDbCommand

        Dim myDA As Data.OleDb.OleDbDataAdapter

        Dim myCB As Data.OleDb.OleDbCommandBuilder

        Dim myDS As DataSet

        Dim myDS1 As DataSet

        Dim myRow As DataRow

        Dim curRow As DataRow

        Dim strconn As String

        Dim strSQL As String

        strconn = "Provider=Microsoft.Jet.OLEDB.4.0;"

        strconn = strconn & "Data Source=C:\Giovanni\FlAnag2000.mdb;"

        strconn = strconn & "Persist Security Info=False;"

        strSQL = "SELECT * FROM FLANAG"

        conn = New Data.OleDb.OleDbConnection(strconn)

        myDS = New System.Data.DataSet()

        myCmd = New Data.OleDb.OleDbCommand(strSQL, conn)

        myDA = New Data.OleDb.OleDbDataAdapter(myCmd)

        myCB = New Data.OleDb.OleDbCommandBuilder(myDA)

        myDA.Fill(myDS, "FLANAG")

        myRow = myDS.Tables(0).NewRow

        myRow("ANAGRAF") = TextBox4.Text

        myRow("INDIRIZZO") = TextBox6.Text

        myRow("DATAN") = CInt(TextBox5.Text)

        myDS.Tables(0).Rows.Add(myRow)

        myDS1 = myDS.GetChanges(DataRowState.Added)

        myDS1.Merge(myDS)

        myDA.Update(myDS, "FLANAG")

        myDS.AcceptChanges()

    End Sub

End Class

1. Nelle variabili strconn e strSQL vengono preparati rispettivamente la stringa di connessione e l’istruzione SQL da eseguire (vedi evento precedente).

2. Viene predisposta la connessione al database e un oggetto Dataset per contenere il recordset su cui eseguire l’aggiornamento.

3. Viene creato un comando (OleDbCommand) che attraverso una connessione a database eseguirà l’istruzione SQL per la selezione del recordset su cui eseguire le modifiche. Nello stesso contesto viene creato il DataAdapter che si occuperà di reperire i dati da database ed eseguire le modifiche sul database. Nel DataAdapter vengono pure create le istruzioni necessarie per l’aggiornamento dei dati sul database usando un CommandBuilder.

4. Attraverso il metodo Fill viene eseguita l’istruzione SQL e viene popolato il Dataset myDS con i dati del recordset da modificare.

5. Il metodo NewRow crea una nuova riga (record) vuota.

6. Vengono popolati, con la solita sintassi i campi del nuovo record aggiunto.

7. La riga viene aggiunta al recordset con il metodo Add di Rows.

8. A questo punto il salvataggio delle modifiche sul database richiede i seguenti passi:

a. Viene creato un dataset contenente solo i record inseriti (uno solo) nel Dataset originale (myDS): metodo GetChanges.

b. Con il metodo Merge i cambiamenti del nuovo dataset vengono inclusi nel dataset originale (myDS).

c. Attraverso i metodi Update dell’oggetto DataAdapter e AcceptChanges del Dataset si trasferiscono effettivamente le modifiche del Dataset sul database.

